



DECUS

PROGRAM LIBRARY

DECUS NO.	FOCAL8-128
TITLE	PROBABILITY (2P); FROM \dagger ("STUDENT") DISTRIBUTION
AUTHOR	Milton Landowne, M.D.
COMPANY	U. S. Army Institute of Environmental Medicine Natick, Massachusetts
DATE	July 1970
SOURCE LANGUAGE	FOCAL

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

PROBABILITY (2P); FROM t ("STUDENT") DISTRIBUTION

DECUS Program Library Write-up

DECUS NO. FOCAL8-128

This program replaces commonly used tables of t providing significance limits of the "Student" distribution. It calculates the probability (2P) that an observed difference between means is due to chance, when given the number of degrees of freedom (n) and the ratio (t) of the difference between means and the standard error of this difference. The program may readily be appended to any statistical program which obtains means, standard error and t; e.g., for paired samples, unpaired samples or differences from zero.

The calculation is based upon the expression for the ordinate of the distribution of t, given by Fisher and Yates (Statistical Methods for Research Workers, 4th Edition, Oliver and Boyd, Edinburgh, 1953, p. 1). Probability (2P) is calculated as the fraction of the frequency falling outside the limits $\pm t$, of the integral of this expression; as presented by Lewis (Quantitative Methods in Psychology, McGraw Hill, New York, 1960, p. 298):

$$2P = 1 - \frac{2 \Gamma[(n+1)/2]}{\sqrt{\pi} \Gamma[n/2]} \int_0^x \cos^{n-1} x \, dx$$

where $t/\sqrt{n} = \tan x$, and the integral is evaluated according to standard tables (Mathematical Tables, 11th Edition, Chemical Rubber Publishing Co., Cleveland, 1960, No. 235, p. 271) as:

$$\int \cos^{n-1} x \, dx = \frac{1}{n-1} \cos^{n-2} x \sin x + \frac{n-2}{n-1} \int \cos^{n-3} x \, dx$$

C-PROBABILITY (2P); FROM T ("STUDENT") DISTRIBUTION

C-FOCAL, 1969

```

30.05  A  ? NU , T ?
30.10  S  X=FATN(FABS(T)/FSQT(NU));S T=0;S Q=1;F I=1,2;S G(I)=1
30.20  I  (NU-2)30.4;S T=FSIN(X)*FEXP((NU-2)*FLOG(FCOS(X)))/(NU-1)
30.30  I  (NU-4)30.4;F K=2,2,NU-2;D 31
30.40  F  I=1,2;F K=1,2,NU-1;S G(I)=G(I)*K/2
30.45  I  (FTR(NU/2)-NU/2)30.5,30.6,30.5
30.50  S  P=1-.63662*(T+Q*X/2)*G(2)/G(1);G 30.7
30.60  S  P=1-2*T*G(1)/G(2)
30.70  T  " 2P ",%6.05,P,!

31.10  S  Q=Q*(NU-K)/(NU-K+1)
31.20  S  T=T+Q*FSIN(X)*FEXP((NU-K-2)*FLOG(FCOS(X)))/(NU-K-1)

F      J=1,7;D 30
NU      :10 , T :1.09307      2P = 0.300000
NU      :10 , T :2.22820      2P = 0.050000
NU      :10 , T :2.85000      2P = 0.01725
NU      :10 , T :3.16930      2P = 0.010000
NU      :10 , T :4.58720      2P = 0.001000
NU      :10 , T :6.20000      2P = 0.000100
NU      :10 , T :8.00000      2P = 0.00001

```

COMMENTS:

- 30.05/ NU is number of degrees of freedom. T is ratio of mean/standard error.
- 30.1 / X is set as angle (in radians) whose tangent is T/\sqrt{NU}
- 30.2 / Unless $NU < 2$, first term of integral is $1/(NU-1)X (\sin X \cos^{NU-2} X)$
- 30.3 / Unless $NU < 4$, succeeding lower ordered terms of integral (except for last term if NU is odd) are calculated and summed by group 31.
- 31.1 / Q is a fractional coefficient calculated as a geometric series expansion for each remaining trigonometric term of integral.
- 31.2 / Summation of integral terms in an arithmetic series, where each term is $Q/(NU-K-1) X (\sin X \cos^{(NU-K-1)} X)$
- 30.40 / Evaluates G1 and G2, geometric series used to calculate factorials for Gamma functions in numerator and denominator of constant term.
- 30.45 / Branches for odd or even values of NU.
- 30.5 / For odd NU; last term is added to sum of integral terms, constant term is ratio $(G1/G2)/\pi$. Probability $(2P)=1-2 \times \text{sum of integral terms} \times \text{constant term}$.
- 30.6 / For even NU; constant term is ratio of $G2/G1$. Probability $(2P)=1-2 \times \text{sum of integral terms} \times \text{constant term}$.
- 30.7 / Types probability (2P)